////////////////////////////////////////////////////////////////////////////////
//
//  The MIT License
//
//  Copyright (c) 2008 - 2009. Dimarik
//
//  Permission is hereby granted, free of charge, to any person obtaining a copy
//  of this software and associated documentation files (the "Software"), to deal
//  in the Software without restriction, including without limitation the rights
//  to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
//  copies of the Software, and to permit persons to whom the Software is
//  furnished to do so, subject to the following conditions:
//
//  The above copyright notice and this permission notice shall be included in
//  all copies or substantial portions of the Software.
//
//  THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
//  IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
//  FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
//  AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
//  LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
//  OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
//  THE SOFTWARE.
//
////////////////////////////////////////////////////////////////////////////////

package ru.goodcore.utils
{

/**
 * @return					XML с описанием вызываемого метода.
 *
 * @author					BlooDHounD
 * @version					1.0
 * @playerversion			Flash 9
 * @langversion				3.0
 */
public function getCallerInfo(nestingLevel:uint = 2):XML {
    var instance:String = ( new Error() ).getStackTrace().match( PATTERN_STACK )[nestingLevel];
    if ( instance ) {
        var result:Array = instance.match(PATTERN_INSTANCE);
        var node:XML;
        if ( result[9] ) {
            node = <node />;
            node.@name = result[9];
            node.setName( result[6] ? "accessor" : "method" );
            if ( result[8] ) {
                node.@uri = result[8];
            }
        }

        var type:XML;
        if ( result[4] && result[4] != "global" ) {
            type = <type />;
            type.@name = result[4];
            if ( result[3] ) {
                type.@uri = result[3];
            }
            if ( node ) {
                type.appendChild( node );
            }
        }

        if ( type ) return type;
        if ( node ) return node;
    }
    return null;
}

}

/**
 * @private
 */
internal const PATTERN_STACK:RegExp = /(?<=^\sat\s).+?(?=\(\))/gm;

/**
 * @private
 */
internal const PATTERN_INSTANCE:RegExp = /^(((.+?)::)?(.+?))?(\/(get\s|set\s)?((.+?)::)?(.+?))?$/;